

## § 42. Використання логічних функцій

**Вивчивши цей параграф, ми:**

дзнаємося про синтаксис та правила застосування логічних функцій;

дзнаємося про операції відношення та логічні операції в Microsoft Excel;

навчимося використовувати логічні функції обчислення простих та складених умов.

### 42.1. Функція розгалуження

Формули, які ми розглядали в попередніх параграфах, виконували обчислення за одним правилом для будь-яких значень початкових даних. Однак часто при виконанні обчислень мають застосовуватися *різні формули* в залежності від цих даних. Наприклад, при розв'язанні квадратного рівняння кількість його коренів та їхні значення залежать від знаку дискримінанта.

В Excel для організації процесу обчислення в таких випадках передбачена **функція розгалуження**, яка має такий синтаксис:

IF(умова; значення1; значення2)

Працює ця функція в такий спосіб: спочатку перевіряється умова; якщо вона виконується, то результатом функції буде значення1, якщо не виконується — значення2.

Умова в Excel — це два вирази, поєднані **операцією відношення**. Часто такі умови називають ще **простими умовами**. В Excel визначені такі операції відношення:

Позначення операції відношення	Назва
=	дорівнює
<>	не дорівнює
>	більше
<	менше
>=	більше або дорівнює
<=	менше або дорівнює

Приклади написання умов:

A1=3

B2<>F4+3

SQRT(X1\*X1+Y1\*Y1) <=R1

Як вирази в умові можуть виступати і числові, і текстові дані. Текст у такому випадку береться в лапки.

Одне із значень у функції IF може бути відсутнім, тоді взамін відсутнього значення1 буде використано значення ІСТИНА, взамін відсутнього значення2 — значення ХИБНІСТЬ. Якщо ви бажаєте, щоб у клітинці в разі виконання (або невиконання) деякої умови не виводилося нічого, то замість значення1 (або значення2 відповідно) треба написати подвійні лапки "".

Функції IF можуть бути вкладеними, тобто другим або третім аргументом однієї функції (значення1 або значення2 відповідно) може бути інша функція IF.

Приклади написання функції IF:

обчислення модуля від числа:

IF(A1>=0;A1;-A1)

порівняння двох чисел:

IF(A1>=B1;"Більше або дорівнює";"Менше")

перевірка, чи дорівнюють 0 три числа одночасно:

IF(A1\*A1+B2\*B2+C3\*C3=0;"Усі три числа дорівнюють 0";)

Перевірка, чи є задане число однозначним, двозначним або більше:

IF(A1<10;"однозначне"; IF(A1<100;"двозначне";"більше 100"))

Наведемо приклад таблиці «Квадратне рівняння» (рис. 42.1).

	A	B	C	D		A	B	C	D
1	Коефіцієнти:		Розв'язання:		1	Коефіцієнти:		Розв'язання:	
2	a=	1	D=	16	2	a=	9	D=	0
3	b=	2	x1=	1	3	b=	6	x1=	-3
4	c=	-3	x2=	-3	4	c=	1	x2=	

а) б)

	A	B	C	D
1	Коефіцієнти:		Розв'язання:	
2	a=	3	D=	-80
3	b=	2	x1=	Немає коренів
4	c=	7	x2=	

в)

Рис. 42.1. Таблиця «Квадратне рівняння»

У цій таблиці введення формул можна виконати в такому порядку:

1. Обчислення дискримінанта квадратного рівняння (клітинка D2):  
 $=B3*B3-4*B2*B4$

2. Корінь  $x1$  (клітинка D3):

$=IF(D2>=0; (-B3+SQRT(D2))/2/B2; "Немає коренів")$

3. Корінь  $x2$  (клітинка D4):

$=IF(D2>0; (-B3-SQRT(D2))/2/B2; "")$

При обчисленні враховуються три варіанти: два корені (рис. 42.1а), один корінь (рис. 42.1б), немає коренів (рис. 42.1в).

Наведемо приклад ще однієї таблиці «Абітурієнт», в якій визначається список зарахованих до вищого навчального закладу абітурієнтів за результатами вступних іспитів (рис. 42.2). Вона складена із застосуванням вкладених логічних функцій.

	A	B	C	D	E	F
1	Середній бал, достатній для зарахування на бюджетне навчання:					4,5
2	Середній бал, достатній для зарахування на контракт:					3
3	Результати іспитів					
		Математика	Історія України	Українська мова	Середній бал	Відмітка про зарахування
4	П.І.Б. учня					
5	Акимов А.П.	5	4	3	4,00	Контракт
6	Бобкін П.В.	2	2	3	2,33	Не зарахований
7	Клакин О.О.	5	5	5	5,00	Бюджет
8	Мощенко Л. О.	4	4	5	4,33	Контракт
9	Нущенко П. П.	5	4	5	4,67	Бюджет
10	Понукович Р. Р.	3	3	4	3,33	Контракт

Рис. 42.2. Таблиця «Абітурієнт»

При створенні цієї таблиці введення формул виконувалося в такому порядку:

1. Середній бал Акімова за результатами іспиту (клітинка E5):

$=AVERAGE(B5:D5)$

2. Середні бали інших абітурієнтів (клітинки E6:E10): через автозаповнення, починаючи з

клітинки E5.

3. Позначка про зарахування Акімова (клітинка F5):

```
=IF (E5>=F$1; "Бюджет";  
    IF (E5>=F$2; "Контракт"; "Не зарахований"))
```

Якщо середній бал Акімова (E5) достатній для зарахування на бюджет ( $\geq F1$ ), то в клітинку виводиться повідомлення «Бюджет». Якщо середній бал не є достатнім для бюджету, то він порівнюється зі значенням F2, після чого в клітинку виводиться одне з двох повідомлень: «Контракт» або «Не зарахований».

4. Позначка про зарахування інших абітурієнтів (клітинки F6:F10): через автозаповнення, починаючи з клітинки F5.

## ====42.2. Складені умови=====

В окремих випадках користувач стикається з необхідністю організації одночасної перевірки двох (або більше) простих умов при виконанні обчислень. Наприклад, при перевірці, чи є значення в клітинці X двозначним, необхідно перевірити дві умови:  $9 < X$  та  $X < 100$ .

Сукупність простих умов називають *складеною умовою*. Для організації складених умов в Excel існують три *логічні функції*, за властивостями вони подібні до логічних операцій у математиці.

1. Логічна функція **AND**.

Функція має такий синтаксис:

```
AND(умова1; умова2; ...)
```

де умова1, умова2, ... — це прості або складені умови. Сукупність цих умов виконується, якщо *кожна* з них виконується.

2. Логічна функція **OR**.

Функція має такий синтаксис:

```
OR(умова1; умова2; ...)
```

де умова1, умова2, ... — це прості або складені умови. Сукупність цих умов виконується, якщо *хоч одна* з них виконується.

3. Логічна функція **NOT**.

Функція має такий синтаксис:

```
NOT(умова)
```

де умова — це проста або складена умова. NOT(умова) виконується, якщо умова не виконується.

Приклади запису складених умов:

```
IF(AND(9<X1; X1<100); "Число двозначне");)
```

```
IF(OR(M1=1; M1=2; M1=12); "Зимовий місяць");)
```

```
IF(NOT(AND(A1<X1; X1<A2; B1<Y1; Y1<B2)); "Точка не належить  
прямокутнику");)
```

	A	B	C	D	E
1	№	Сторони			Чи є
2	з.п.	a	b	c	рівнобічним?
3	1	2	3	4	
4	2	2	4	4	рівнобічний
5	3	5	4	7	
6	4	5	3	5	рівнобічний
7	5	3	7	8	

Рис. 42.3. Таблиця «Рівнобічний трикутник»

Наведемо приклад таблиці «Рівнобічний трикутник», яка створена для визначення, чи є трикутники із заданими довжинами сторін рівнобічними (рис. 42.3). Відомо, що трикутник називається рівнобічним, якщо в нього дві сторони дорівнюють одна одній.

При створенні цієї таблиці введення формул виконувалось у такому порядку:

1. Визначення типу першого трикутника (клітинка E3):

=IF (OR (B3=C3; B3=D3; C3=D3) ; "рівнобічний" ; "")

2. Визначення типів інших трикутників (клітинки E4:E7): через автозаповнення, починаючи з клітинки E3.

## ВИСНОВКИ

Функція розгалуження IF використовується у випадках, коли значення в клітинці має обчислюватись за різними формулами в залежності від виконання певної умови. Першим аргументом функції IF є умова. Якщо умова виконується, то результатом функції буде значення другого аргументу, якщо не виконується — третього. Прості умови в MS Excel створюються за допомогою однієї з операцій відношення, які доз-воляють порівнювати значення двох виразів. При необхідності перевірки виконання декількох умов використовуються логічні функції, вони подібні до логічних операцій у математиці. Декілька простих умов, поєднаних логічними функціями, утворюють складену умову.

## Контрольні питання та вправи

1. Укажіть, які з наведених формул Excel записані без синтаксичних помилок:

а) =IF (B1<C1) THEN (B1)

б) =IF (B1<C1) THEN (B1) ELSE (C1)

в) =IF (B1<C1; B1)

г) =IF (B1<C1; B1; C1)

2. Укажіть, яке повідомлення з'явиться в клітинці в результаті обчислення формули =IF (A1^2+B1^2<>25; "не належить колу"; 5), якщо значення в клітинках A1 та B1 дорівнюють 4 та 5 відповідно:

а) A1^2+B1^2<>25

б) =IF (A1^2+B1^2<>25; "не належить колу"; 5)

в) не належить колу

г) 5

3. Визначте, які з наведених тверджень є правильними:

а) AND (умова1; умова2; ...), де умова1, умова2, ... — це прості або складені умови. Сукупність цих умов виконується, якщо будь-яка з них виконується.

б) OR (умова1; умова2; ...), де умова1, умова2, ... — це прості або складені умови. Сукупність цих умов виконується, якщо більшість із них виконується.

в) NOT (умова1; умова2; ...), де умова1, умова2, ... — це прості або складені умови. Сукупність цих умов виконується, якщо хоч одна з них виконується.

г) IF (умова; значення1; значення2). Якщо умова виконується, то значенням функції буде значення1, в іншому випадку — значення2.

4. Укажіть, які з наведених складених умов виконуються:

а) AND (2<3; 4<>5)

б) OR (6=45/5; 4<=5; 45>=45)

в) NOT (OR (5<>4; 6<>6))

г) OR (AND (67=89; 6>=5^2); AND (5=6\*2; 7^2=2^7))

5. Опишіть синтаксис та правила використання функції розгалуження IF. Наведіть приклади її використання.

6. Наведіть перелік відомих вам операцій відношення та приклади їх використання. Запишіть умови, які перевіряють, чи є значення клітинки A1:

1) тризначним;

2) від'ємним;

3) не рівним нулю.

7. Наведіть перелік відомих вам логічних функцій, що призначені для організації складених умов. Опишіть одну з цих функцій.

8. Напишіть складені умови визначення:

- 1) чи є заданий чотирикутник паралелограмом, якщо задано довжини його сторін у клітинках A1:A4;
- 2) чи є заданий чотирикутник прямокутником, якщо задані координати його вершин у клітинках X1:Y4.

9. За допомогою табличного процесора Microsoft Excel створіть таблицю перевірки плавучості фізичних речовин. Таблиця має містити три стовпці: «Назва речовини», «Густота», «Здатність плавати». Перші два стовпці заповніть самостійно за таблицями з підручника фізики, третій має заповнитися автоматично за допомогою формули, яка порівнює густоту тіла з густотою води.

10. За допомогою табличного процесора Microsoft Excel створіть таблицю розв'язання лінійного рівняння  $ax + b = 0$ .

11. За допомогою табличного процесора Microsoft Excel створіть таблицю-тест для перевірки знань з інформатики. Таблиця має містити 12 запитань, на кожне з яких можна відповісти «так» чи «ні». Після надання відповідей на запитання автоматично заповнюється стовпець «Бали» (1 бал — за правильну відповідь і 0 — за неправильну). Оцінка за розв'язання тесту дорівнює сумі отриманих балів.

Приклад таблиці, що складається з трьох запитань до виконання тесту:

№	Запитання	Відповідь	Бали
1	Аркуш Microsoft Excel складається з 65535 рядків?		
2	Чи можна за допомогою автозаповнення створити геометричну прогресію?		
3	(умова1) ИЛИ (умова2) Функція ИЛИ має такий синтаксис?		
Оцінка			

Приклад таблиці після виконання тесту:

№	Запитання	Відповідь	Бали
1	Аркуш Microsoft Excel складається з 65535 рядків?	так	0
2	За допомогою автозаповнення можна створити геометричну прогресію?	так	1
3	(умова1) ИЛИ (умова2) Функція ИЛИ має такий синтаксис?	ні	1
Оцінка			2

12. За допомогою Microsoft Excel створіть таблицю «Трикутник», яка за даними сторонами трикутника визначає тип трикутника: різнобічний / рівнобічний / рівносторонній та тупокутний / гострокутний.